DOCUMENT RESUME

ED 174 465 SE 028 537

AUTHOR	Dalton, Joel B.
TITLE	Orbital Mechanics.
INSTITUTION	Dartmouth Coll., Hanover, N.H. Kiewit Computation Center.
SPONS AGENCY	National Science Foundation, Washington, D.C.
PUB DATE	Feb 70
GRANT	NSF-GW-2246
NOTE	33p.; For related documents, see SE 028 535-538; Not available in hard copy due to copyright restrictions; Computer print-out will not reproduce well
EDRS PRICE	MF01 Plus Postage. PC Not Available from EDRS.
DESCRIPTORS	*Computer Oriented Programs; *Computer Programs; Computers; *Mathematics Education; *Physics; Physics Curriculum; *Science Education; Secondary Education; Simulation

ABSTRACT

Three computer programs are presented that allow the high school student to explore and understand the physical forces involved in orbital flight at a greater depth than is usually possible. For each program, introductory material is given including the physics and mathematics involved. This is followed by the computer program in BASIC language. The mathematics involved is essentially that of an advanced algebra II course. The three program topics are elliptic orbits, rendezvous, and orbital transfer. (MP)

SE
Ki

ORBITAL MECHANICS

by

Joel B. Dalton
Hanover High School

SECONDARY SCHOOL PUBLICATION

KIEWIT COMPUTATION CENTER
DARTMOUTH COLLEGE
HANOVER, NEW HAMPSHIRE 03755

(Supported in part by National Science Foundation Grant No. GW-2246)

February 1970

TABLE OF CONTENTS

INTRODUCTION

The original planning and experimentation from which these programs evolved was done in a Math/Physics seminar at Hanover High School in the Spring of 1969. We wished to explore and understand the physical forces involved in orbital flight at a depth for which no secondary school literature appeared to exist. The DTSS* presented an interesting opportunity for the students to do actual constructive research using mathematical models and computer simulations.

The mathematics is essentially that of a superior Algebra II course that stresses Trigonometry and Analytic Geometry. A student's ability in these areas may be sorely tried and perhaps extended if these programs and supporting material are dissected and examined. Many problem solutions normally obtainable only through the methods of calculus are found through computer programs utilizing some interesting algorithms. The original questions asked by the seminar students were non-trivial, and the resulting analyses were, consequently, slightly more sophisticated than was expected. Despite this creeping complexity, the questions that were answered at all were answered with a substantial degree of integrity.

The physics represents an extension of the basic physics employed in the excellent Holt, Rinehart & Winston paperback series on space flight with specific emphasis on "Mathematics of Space Flight", the most utilitarian of these paperbacks. Obviously, the programs go beyond this introductory material but may still be considered a linear extension of it. Vectors and energy levels are the prime tools of kinetics analysis. At least one seminar of a good physics course such as PSSC is heartily recommended as a prerequisite to this material although much of the necessary background is well introduced by the paperback text.

The forces and energies studied have such a large magnitude and many of the changes sought after are so minute that the computer cannot give the accuracy of computation that some students would desire. However, the numerical analyses yield data that is sufficient for most uses.

During the Spring of 1970, some Hanover students are planning to write exercise sets around the present programs, and perhaps, some new programs as well. These exercises will lie in three categories: (1) questions requiring extensive knowledge of the given program and its accompanying Topic Outline and also requiring many directed runs of the present program; (2) questions

* Dartmouth Time-Sharing System

requiring minor re-writing of the program to present data from a different point of view; and (3) compilation of tables to aid in analysis directed toward writing new exploratory programs and requiring continuous and systematic re-programming. These exercise sets may be available in a later edition of this topic outline. The following program notes may help in creating ad hoc exercises.

(1) Elliptic Orbits: This topic is a time analysis of closed orbits. The time analysis is made possible through the use of a computer algorithm that rather accurately replaces the commonly used elliptic integrals. The laws of planetary motion may be studied quite closely after minor substitutions are made in the program. Keppler's laws, as one example, are directly observable on a time plot.

(2) Rendezvous: The simplest form of the two ship rendezvous is solved and the time function is used again to track the interception in either tabular or graphic displays. The program is stripped to the absolute minimum to emphasize "window" and relative orbital constants. Obviously, a reversal of the ship roles in the program leads to the Hohman transfer. The program as it stands gives a clear picture of circular and elliptic orbits of the same magnitude.

(3) Orbital Transfer: This program computers the effect of adding any energy vector to any orbit configuration at any point. The potential of this program as yet is largely unexplored and its limitations and weaknesses have not yet been fully identified. Suggested exercises would include studies of orbital stability, relative influences of the multiple variables, and of course, problems in rotation of orbital axes. The advanced student may use an easy variation of the program as a flight simulator in exchanging orbits.

Future developments will hopefully extend Rendezvous to include all manner of orbits. This will, of course, require the most careful exploitation of Orbital Transfer to control rotation and alignment of orbital axes.

I. ELLIPTIC ORBITS

REF: ELORB

Time analysis of closed orbits has been a subject of fascination to astronomers, physicists, and mathematicians alike for some 300 years. Theoretically, if the orbit is known, then the orbiting body's position is strictly a function of time. Unfortunately for the untutored amateur or elementary student of orbital mechanics this position plot as a function of time gives rise to some complicated applied mathematics.

The advent of the high-speed computer offers an alternative to the elliptic calculus developed by the giants of the 18th and 19th centuries to handle "celestial mechanics." The ability of a computer to handle large numbers of complex calculations in fractions of a second allows "back solving" of systems of equations which, if solved in a straightforward manner, lie in the domain of advanced mathematics.

Before examination of the actual equations from which the program is constructed, a look at the physical situation is convenient. In Figure 1 is shown a typical elliptic orbit with Earth at its principle focus. A space ship is shown at an instant of time along with the area which its radius vector has swept out.



Figure 1.

Since an orbit which is undisturbed by additions or subtractions of energy has constants for its descriptive parameters, it is obvious that a few astronomical observations will yield:

1. Closest approach to Earth

2. Furthest recession from Earth

From these data all other significant parameters may be calculated. From an arbitrary zero time it seems as though after K seconds a given area is swept out by the radius vector which is directly dependent on θ, the radius vector angle.

Keppler established that the radius vector sweeps out equal areas in equal times. If it were easily calculable, the area of an ellipse in terms of θ would be of immense aid since we could then directly relate t and θ. In Figure 2 such a calculation is established. The result is an integral which is included in the text as a "given" equation but should be briefly examined at this point.



Figure 2. The Differential Triangular Area of an Ellipse

From Figure 2 the area of $\triangle$ AOB is $\frac{1}{2}r.r.d\theta$, and the integral (where $r = L/1{+}e\cos\theta$) is

$$A\triangle \Big|_a^b = \frac{L^2}{2} \int_a^b \frac{d\theta}{(1 + e\cos\theta)^2}$$

which is a standard transcendental with a solution readily available from any set of integral tables.

In most mathematical texts the solution to the area problem is manipulated so that $t = f(\theta)$ which is, of course, the absolute inverse to what is required, i.e., $\theta = f(t)$. It is precisely at this point that a secondary school student with a computer can match the calculations of the more advanced student.

The program associated with the mathematical portion of this topic outline can be set to report angular position (r, θ) given any time interval. This is accomplished by having the computer search for a θ that will produce the required time. The present program is adjusted for 5 minute intervals with an accuracy of .001 minutes, or 15 search sweeps, whichever occurs first. This avoids the inverse solution to the differential triangular area of the ellipse which leads to elliptic integrals which would have to be solved for each time plot.

When "approach" and "recession" distances are given to the computer in response to its input interrogation, the program not only presents most of the orbital data in tabular form but on command will track the orbiting object giving position plots with velocities at 5 minute time intervals.

As a consequence, any student with an elementary background in mathematics who gain access to a time-shared computer, may study closed orbits with their associated laws of physical behavior. The more serious student who cares to pursue the mathematics of this physical behavior will find all the equations necessary for further exploration and possible re-programming.

Objects in free fall around Earth fall in orbits which are mathematically describably as conic sections. The principal focus is at Earth center. The polar coordinate system is used to describe a ship's orbital position. The reference axis is through Earth center on the orbital perigee. The ship's position in an elliptic (as well as any other conic orbit) is then

$$r = \frac{L}{1 + e \cos \theta} \qquad \text{(E1)}$$



Figure 3. Orbital Parameters (Elliptic Orbit)

A: Apogee
P: Perigee
L: Semi Latus Rectum
a: Semi Major Axis
b: Semi Minor Axis
ae: Semi Focal Length

r: Radius vector
θ: Radius angle
e: Eccentricity

NOTE: Counterclockwise orbit is with Earth rotation and will be standard.

The distance parameters in terms ot A, P are:

$$a = (A + P)/2 \qquad \text{(E2)}$$

$$e = (A - P)/(A + P) \qquad \text{(E3)}$$

$$L = a(1 - e^2) \qquad \text{(E4)}$$

$$b = a\sqrt{1 - e^2}$$

The inertial period, Pe, of an elliptic orbit is

$$Pe = 2\pi \sqrt{\frac{a^3}{Gm}} \quad \text{seconds} \tag{E6}$$

where Gm, the Earth gravitational constant is

$$Gm = 62747 \text{ nautical miles/second}^2. \tag{E7}$$

The ship's velocity at any point in orbit is

$$v = \sqrt{Gm\left(\frac{2}{r} - \frac{1}{a}\right)} \text{ nmps.} \tag{E8}$$

Velocities at apogee and perigee will be

$$v_P = \sqrt{\frac{Gm}{A}(1+e)} \text{ nmps} \tag{E9}$$

$$v_A = \sqrt{\frac{Gm}{P}(1-e)} \text{ nmps.} \tag{E10}$$

And the area swept out by the radius vector is

$$A(r,\theta) = \frac{L^2}{2}\left[\frac{e \sin\theta}{(e^2-1)(1+e\cos\theta)} + \frac{2}{(1-e^2)^{3/2}} \text{Tan}^{-1}\left(\sqrt{1-e^2}\ \text{Tan}\frac{\theta}{2}\right)\right] \tag{E11}$$

By Keppler's 2nd Law, the following relationship holds:

$$\frac{A(r,\theta)}{k} = t \text{ seconds} \tag{E12}$$

If (E11) is solved for one complete revolution

$$A(r,\theta) = \pi ab \tag{E13}$$

and the following equalities may be established:

$$\pi ab = 2\pi \frac{a^3}{Gm} \quad k \tag{E14}$$

$$t = \frac{L^2\sqrt{aGm}}{60bGm}\left[\frac{e\cos\theta}{(e^2-1)(1+e\cos\theta)} + \frac{2}{(1-e^2)^{3/2}} \text{Tan}^{-1}\left(\sqrt{1-e^2}\ \text{Tan}\frac{\theta}{2}\right)\right] \text{minutes} \tag{E15}$$

To directly solve (E15) for θ in terms of t is evidently quite difficult, so the time function is computed through a computer convergence or search program.

```
ELØRB     03/18/70  09:51

THIS PRØGRAM IS KEYED TØ THE TØPIC ØUTLINE 'ELLIPTIC
ØRBITS' THRØUGH THE REM CØLUMN REFERENCES

CLØSEST APPRØACH TØ EARTH? 150
FURTHEST RECESSIØN FRØM EARTH? 950


CØNSTANT PARAMETERS
-------- ----------

SEMI MAJØR AXIS IS 3992
SEMI MINØR AXIS IS 3971.91
SEMI LATUS RECTUM IS 3951.92
SEMI FØCAL LENGTH IS 400.
ECCENTRICITY IS 0.1002
INERTIAL PERIØD IS 105.443
PERIGEE VELØCITY IS 15782.2
APØGEE VELØCITY IS 12907.5


VARIABLE PARAMETERS
-------- ----------

ELAPSED TIME    RADIUS ANGLE     RADIUS V        VELØCITY
------- ----    ------ -----     ------ -        --------

 0               1.09863 E-2      3592.           15782.2
 5               18.8525          3609.64         15711.9
 10              37.5403          3661.04         15509.3
 15              55.9314          3741.88         15196.7
 20              73.905           3845.11         14807.8
 25              91.3733          3961.43         14382.3
 30              108.292          4080.24         13960.5
 35              124.706          4191.02         13573.
 40              140.68           4284.           13264.2
 45              156.281          4351.07         13041.7
 50              171.661          4386.84         12924.3
 55              186.976          4388.38         12919.3
 60              202.346          4355.58         13026.9
 65              217.936          4291.03         13240.7
 70              233.855          4200.15         13546.8
 75              250.225          4090.6          13924.4
 80              267.111          3971.98         14344.4
 85              284.535          3854.98         14771.2
 90              302.465          3750.21         15164.9
 95              320.812          3667.12         15485.5
 100             339.489          3612.86         15699.1
 105             358.33           3592.14         15781.6

TIME: 2.000 SEC.
```

ELØRB

```
100 PRINT "THIS PRØGRAM IS KEYED TØ THE TØPIC ØUTLINE 'ELLIPTIC"
110 PRINT "ØRBITS' THRØUGH THE REM CØLUMN REFERENCES"
120 PRINT
130 PRINT "CLØSEST APPRØACH TØ EARTH";
140 INPUT X
150 LET X = X+3442                                         'P
160 PRINT "FURTHEST RECESSIØN FRØM EARTH";
170 INPUT Y
180 LET Y = Y+3442                                         'A
190 LET A = (X+Y)/2                                        '(E2)
200 LET E=(Y-X)/(Y+X)
202 IF E<1 THEN 210
204 PRINT "ECCENTRICITY ØF SELECTED ØRBIT IS"E
206 PRINT "REDUCE ECCENTRICITY BELØW E=1"
208 GØTØ 120
210 LET L = A*(1-E↑2)                                      '(E4)
220 LET B = A*SQR(L/A)                                     '(E5)
230 LET P = 2*3.14159265*SQR(A↑3/62747)                    '(E6)
240 LET V(2)= SQR(62747/Y)*SQR(1-E)
250 LET V(1) = SQR(62747/X)*SQR(1+E)
260 PRINT
270 PRINT
280 PRINT "CØNSTANT PARAMETERS"
290 PRINT "-------- ----------"
300 PRINT
310 PRINT "SEMI MAJØR AXIS IS"A
320 PRINT "SEMI MINØR AXIS IS"B
330 PRINT "SEMI LATUS RECTUM IS"L
340 PRINT"SEMI FØCAL LENGTH IS"A*E
350 PRINT"ECCENTRICITY IS"E
360 PRINT "INERTIAL PERIØD IS" P/60
370 PRINT "PERIGEE VELØCITY IS" V(1)*3600
380 PRINT "APØGEE VELØCITY IS" V(2)*3600
390 PRINT
400 PRINT
410 PRINT "VARIABLE PARAMETERS"
420 PRINT "-------- ----------"
430 PRINT
432 PRINT "ELAPSED TIME","RADIUS ANGLE","RADIUS V","VELØCITY"
433 PRINT "------- ----","------ -----","------ -","--------"
435 PRINT
440 LET N = INT(P/60)
450 FØR M = 0 TØ N STEP 5                                  'SEARCH LØØP
455 LET S=0
460 LET Ø = 2*3.14159265                                   'ØVER
470 LET U = 0                                              'UNDER
480 LET Z = (Ø+U)/2                                        'TRIAL ANGLE
490 LET T(1) = L↑2*SQR(A*62747)/(60*B*62747)               '(E15)
500 LET T(2) = E*SIN(Z)/((E↑2-1)*(1+E*CØS(Z)))             '(E15)
510 LET T(3) = 2/SQR(1-E↑2)↑3                              '(E15)
```

ELØRB (CØNTINUED)

```
520 LET T(4) = ATN(SQR(1-E↑2)*TAN(Z/2))                    '(E15)
525 LET S=S+1
530 LET T = T(1)*(T(2)+(T(3)*T(4)))                        '(E15)
532 IF Z <= 3.14159265 THEN 540
535 LET T=T+2*3.14159265*SQR(A↑3/62747)/60
540 IF ABS(T-M) < .001 THEN 600                            'EXIT
545 IF S=15 THEN 600
550 IF T > M THEN 580
560 LET U = Z                                              'NEWUNDER
570 GØTØ 480
580 LET Ø = Z                                              'NEW ØVER
590 GØTØ 480
600 LET R=L/(1+E*CØS(Z))
610 LET V = SQR ((2*A*62747-R*62747)/(A*R))                'VEL.
620 PRINT M,Z*180/3.14159265,R,V*3600
630 NEXT M                                                 'END LØØP
700 END
```

II. RENDEZVOUS

REF: CHASE

The rendezvous problem has been simplified to the minimum necessary to illustrate problems of pursuit. Essentially different orbits have different velocities, and movement between them requires careful calculation if interception is the goal. The following figures illustrate the lead problem and its theoretical solution.



Figure 1.

If the descent is from a co-radial position, the chase ship will be behind the target due to the difference in velocities and times between the descent orbit and the target orbit.



Figure 2.

The chase ship initiates its maneuvers from a calculated lead interval based on co-solving the three orbits for elapsed time and average velocities.

The descent orbit is interesting in its velocity changes. Initially, the chase ship sheds velocity as the elliptic orbit, with its apogee at the chase ship orbit, has a lower energy level at that point. As the ship descends with the force of gravity to its perigee at the target orbit, its velocity increases to the necessary energy level to "crack the whip" at perigee. So once again, velocity must be reduced. In essence, velocity is twice reduced, and in nearly the same amounts, to enter a faster orbit from a slower orbit.

When the program is initiated the original ship orbits are randomized within certain bounds before being presented to the student. A Firing Table is generated which presents all the necessary navigational and maneuvering data for a rendezvous. If the student wishes to track the rendezvous, data is presented in intervals of the student's choice. The search program from ELORB is incorporated into the descent orbit computations and certain other data is computed from this to show the following:

1. Elapsed time in minutes

2. Relative distance between ships in nautical miles

3. Relative angle between the ships (in degrees)

4. Relative velocity between the ships in nautical miles per hour

5. Time in minutes until rendezvous

One of the objectives of this program is for the student to become familiar enough with the program - physics and mathematics - to reverse the positions of the two ships.

Suppose two ships are in differing circular orbits and a rendezvous is desired. Suppose the target ship is in orbit of radius r_2 and the chase ship is in orbit of radius r_1 and further more, the chase ship is displaced L radians from the target and is in the larger (and slower) orbit. The chase ship must now establish a transfer orbit with its apogee on the chase ship original orbit and the perigee on the target ship's orbit.

To describe the transfer orbit we have



Figure 3.

$$A = r_1 \tag{E1}$$

$$P = r_2 \tag{E2}$$

$$e = A - P/A + P \tag{E3}$$

$$S = A + P/2$$

The velocity of the transfer orbit at apogee is

$$v_a = \sqrt{\frac{Gm}{r_1}(1 - e)} \tag{E4}$$

Hence, the amount of velocity to be lost by retrofire is the difference between the velocity of the chase ship's orbit and (E4).

$$x_1 = \sqrt{\frac{Gm}{r_1}} - \sqrt{\frac{Gm(1 - e)}{r_1}} \tag{E5}$$

The time to perigee in the transfer orbit is half the period of the new orbit or

$$t_1 = \pi\sqrt{\frac{S^3}{Gm}} \qquad \text{(E6)}$$

At perigee the chase ship has a velocity higher than its tangential circular orbit and must again reduce velocity to match the target orbit.

$$x_2 = \sqrt{\frac{Gm(1+e)}{r_2}} - \sqrt{\frac{Gm}{r_2}} \qquad \text{(E7)}$$

If, for the purposes of simplicity, only the transfer time is considered, the chase ship must fire its motors precisely t_1 seconds before the target ship reaches the anticipated perigee rendezvous.

Since the target ship travels 2 radians in

$$t = 2\pi\sqrt{\frac{r_2{}^3}{Gm}} \quad \text{seconds,} \qquad \text{(E8)}$$

then its angular velocity may be considered

$$v_4 = \frac{1}{\sqrt{\frac{r_2{}^3}{Gm}}} \quad \text{rad/sec.} \qquad \text{(E9)}$$

Hence, it must travel w_t radians to rendezvous

$$w_t = \text{(E6)(E9)} = \pi\sqrt{\frac{S^3}{r^3}} \qquad \text{(E10)}$$

The chase ship must be radians from the perigee rendezvous so the chase ship's lead at time of firing must be

$$w = \pi\left[\sqrt{\frac{a^3}{r_2{}^3}} - 1\right] \quad \text{radians.} \qquad \text{(E11)}$$

The relative angular velocity of the ships is

$$v = \frac{1}{\sqrt{\frac{r_2{}^3}{Gm}}} - \frac{1}{\sqrt{\frac{r_1{}^3}{Gm}}} \quad \text{rad/sec.} \qquad \text{(E12)}$$

The amount of delay necessary for the relative ship velocities to change the relative ship angle to the proper "window" is found from the two possible cases in Figure 4.



Case I　　　　　　　　　　　　　　　　　　　Case II

Figure 4.

In Case I the target ship is outside the window and

$$t_2 = (2\pi - w_2 - w)/v. \qquad \text{(E13)}$$

In Case II the target ship is inside the window and

$$t_2 = \left[(2\pi - w_2) + (2\pi - w)\right]/v. \qquad \text{(E14)}$$

The argument for a rendezvous with the chase ship occupying the inner position is similar.

CHASE 03/18/70 09:35

FIRING TABLE
------ -----

CHASE SHIP ANGLE IS 0
CHASE SHIP RADIUS IS 4993.67
TARGET SHIP ANGLE IS 254.557
TARGET SHIP RADIUS IS 4088.47

WINDØW ANGLE IS 30.7019
TIME TØ WINDØW IS 87.5434
REDUCE VELØCITY BY 652.627
DESCENT TIME IS 63.9645
REDUCE VELØCITY BY 686.131

DØ YØU WISH TØ TRACK THE SHIPS? YES
TRACKING INTERVAL? 10

TRACKING DATA
-------- ----

ELAPSED TIME	RELATIVE ANGLE	RELATIVE DISTANCE	RELATIVE VELØCITY	TIME TØ RENDEZVØUS
0	105.443	7247.42	1342.1	151.508
10	96.9051	6823.6	1342.1	141.508
20	88.3676	6363.1	1342.1	131.508
30	79.83	5868.77	1342.1	121.508
40	71.2924	5343.75	1342.1	111.508
50	62.7549	4791.56	1342.1	101.508
60	54.2173	4216.25	1342.1	91.508
70	45.6797	3622.66	1342.1	81.508
80	37.1422	3017.08	1342.1	71.508
87.5434	30.7019	2557.88	1342.1	63.9645

EXECUTE INITIAL RETRØFIRE

92.5434	26.888	2279.87	1958.61	58.9645
97.5434	23.1179	1992.76	1852.14	53.9645
102.543	19.5237	1707.75	1679.49	48.9645
107.543	16.1492	1430.23	1448.53	43.9645
112.543	13.0823	1168.47	1169.55	38.9645
117.543	10.367	927.94	856.572	33.9645
122.543	7.95924	709.123	528.478	28.9645
127.543	5.94703	520.315	203.014	23.9645
132.543	4.28638	362.216	98.1257	18.9645
137.543	2.88941	232.557	353.32	13.9645
142.543	1.7561	133.147	545.032	8.96452
147.543	0.75464	54.5755	658.103	3.96452

EXECUTE FINAL RETRØFIRE

151.508	2.19522 E-2	1.56645	2.14577 E-4	5.08502 E-7

TIME: 1.651 SEC.
READY

CHASE

```
100 RANDØMIZE                                         'ØRIGINATE
105 LET J = 180/3.14159265
110 LET R(1) = RND*1000+4000
120 LET R(2) = RND*1000+4000
130 LET W(1) = RND*6.2831853
140 LET W(2) = RND*6.2831853
150 LET V(1) = SQR(62747/R(1))
160 LET V(2) = SQR(62747/R(2))
170 IF R(1) > R(2) THEN 200                           'GATE
180 IF R(1) < R(2) THEN 100
190 GØTØ 100
200 LET A = R(1)                                      'APØGEE
210 LET P = R(2)                                      'PERIGEE
220 LET E = (A-P)/(A+P)                               'ECCENT
230 LET S = (A+P)/2                                   'SEMI MAJ
240 LET T(1) = 3.14159265*SQR(S↑3/62747)              'CØAST TIME
250 LET W = 3.14159265*SQR(S↑3/R(2)↑3)-3.14159265     'WINDØW
260 LET V(3)=1/SQR(R(1)↑3/62747)
270 LET V(4)=1/SQR(R(2)↑3/62747)
280 LET V = V(4)-V(3)                                 'REL
290 LET Z = 6.2831853-W(1)                            'INITIALIZE
300 LET W(1) = 0                                      'ØRBITS
310 LET W(2) = W(2)+Z
320 IF W(2) < 6.2831853 THEN 340
330 LET W(2) = W(2)-6.2831853
340 IF W(2) < (6.28318530-W) THEN 380
350 LET Y = (6.28318530-W(2))+(6.28318530-W)          'TARGET ADV.
360 LET T(2) = Y/V                                    'TIME TØ FIRE
365 IF T(2)/60>100 THEN 100
370 GØTØ 400                                          'EXIT
380 LET Y = (6.28318530-W(2))-W                       'TARGET ADV.
390 GØTØ 360
400 LET X(1) = SQR(62747/A)-SQR(62747*(1-E)/A)        'EXCESS
410 LET X(2) = SQR(62747*(1+E)/P)-SQR(62747/P)            'VELØCITIES
420 PRINT ,,"FIRING TABLE"
430 PRINT ,,"------ -----"
440 PRINT
450 PRINT "CHASE SHIP ANGLE IS"W(1)*180/3.14159265
460 PRINT "CHASE SHIP RADIUS IS"R(1)
470 PRINT "TARGET SHIP ANGLE IS"W(2)*180/3.14159265
480 PRINT "TARGET SHIP RADIUS IS"R(2)
490 PRINT
500 PRINT "WINDØW ANGLE IS" W*180/3.14159265
510 PRINT "TIME TØ WINDØW IS"T(2)/60
520 PRINT "REDUCE VELØCITY BY"X(1)*3600
530 PRINT "DESCENT TIME IS"T(1)/60
540 PRINT "REDUCE VELØCITY BY"X(2)*3600
550 PRINT
560 PRINT "-----------------------------------------------------------"
570 PRINT
```

CHASE (CONTINUED)

```
580 PRINT "DØ YØU WISH TØ TRACK THE SHIPS";
590 INPUT A$
600 IF A$="YES"THEN 620
610 STØP
620 PRINT "TRACKING INTERVAL";
630 INPUT J
640 PRINT
650 PRINT
660 PRINT ,,"TRACKING DATA"
670 PRINT ,,"-------- ----"
680 PRINT
690 PRINT "ELAPSED","RELATIVE","RELATIVE","RELATIVE","TIME TØ"
700 PRINT "TIME","ANGLE","DISTANCE","VELØCITY","RENDEZVØUS"
710 PRINT "----","-----","--------","--------","----------"
720 FØR T = 0 TØ INT(T(2)/60) STEP J
730 PRINT T,
740 LET W(3) = T*60*V(3)+W(1)
750 LET W(4)=T*60*V(4)+W(2)
760 LET M = ABS(W(3)-W(4))
770 IF M < 3.14159265 THEN 790
780 LET M = ABS(M-6.28318530)
790 PRINT M*180/3.14159265,
800 LET D = SQR(A↑2+P↑2-2*A*P*CØS(M))
810 PRINT D,
820 LET V(5) = ABS(V(1)-V(2))
830 PRINT V(5)*3600,
840 PRINT -T+(T(1)+T(2))/60
850 IF T = T(2)/60 THEN 890
860 NEXT T
870 LET T = T(2)/60
880 GØTØ 730
890 PRINT
900 PRINT "EXECUTE INITIAL RETRØFIRE"
910 PRINT
920 LET L = S*(1-E↑2)
930 LET B = S*SQR(L/S)
940 LET W(1) = W(3)
950 LET W(2) = W(4)
960 IF W(1) < 6.2831853 THEN 990
970 LET W(1) = W(1)-6.28318530
980 GØTØ 960
990 IF W(2)< 6.2831853 THEN 1015
1000 LET W(2) = W(2)-6.28318530
1010 GØTØ 990
1015LET J=5
1020 FØR T=(J+T(1)/60) TØ T(1)/30 STEP J
1022 LETQ=Q+1
1030 LET K = 0
1040 LET Ø = 6.2831853
1050 LET U = 0
```

CHASE (CONTINUED)

```
1060 LET Z = (0+U)/2
1070 LET N(1) = L↑2*SQR(S*62747)/(60*B*62747)
1080 LET N(2)=E*SIN(Z)/((E↑2-1)*(1+E*COS(Z)))
1090 LET N(3) = 2/SQR(1-E↑2)↑3
1100 LET N(4) = ATN(SQR(1-E↑2)*TAN(Z/2))
1110 LET N = N(1)*N(2)+N(1)*N(3)*N(4)
1113 IF Z<=3.14159265 THEN1120
1115 LET N=N+T(1)/30
1120 LET K = K+1
1130 IF ABS(N-T)<.01 THEN 1195
1140 IF K=15 THEN 1195
1150 IF N > T THEN 1180
1160 LET U = Z
1170 GOTO 1060
1180 LET 0 = Z
1190 GOTO 1060
1195 IF T<T(1)/30 THEN 1200
1197 LET T(3)=T(1)/60+T(2)/60
1198 GO TO 1205
1200 LET T(3)=Q*J+T(2)/60
1205 PRINT T(3),
1210 LET W(2) = J*60*V(4)+W(2)
1220 LET W=Z-3.14159265 +W(1)
1230 LET M=ABS(W-W(2))
1280 PRINT M*180/3.14159265,
1290 LET R = L/(1+E*COS(Z))
1300 LET D=SQR(R↑2+P↑2-2*R*P*COS(M))
1310 PRINT D,
1315 IF T<T(1)/30 THEN 1320
1316 LET V(1)=SQR((2*S*62747-R*62747)/(S*R))-X(2)
1317 GO TO 1330
1320 LET V(1) = SQR((2*S*62747-R*62747)/(S*R))
1330 LET V(5) = ABS(V(1)-V(2))
1340 PRINT V(5)*3600,
1350 LET T(4)=T(1)/60 +T(2)/60 -T(3)
1351 PRINT T(4)
1360 IF T=T(1)/30 THEN 1420
1370 NEXT T
1380 PRINT
1390 PRINT "EXECUTE FINAL RETROFIRE"
1400 PRINT
1405 LET T=T(1)/30
1406 LET J=T(4)
1407 LET R=P
1410 GO TO 1022
1420 STOP
1430 END
```

ORBITAL TRANSFER

REF: XFR

The problem of "where are we?" besets every pilot of a spaceship once he has fired his ship's motors. The problem in its most simple form is finding the new orbit attained from the old orbit by introducing an energy vector represented in this program by the coplanar thrust of the motors.

The program despite its complexity does have simple inputs and outputs. The inputs for the old orbit are in response to interrogation.

1. Eccentricity of the old orbit.

2. Length of the Semi-Latus Rectum in nautical miles.

3. Position of the ship in degrees from perigee axis.

The program then invites the student to fire the motors which he does by answering further interrogation.

1. Angle at which motors are fired (in degrees).

2. Length of time motors are to be fired (in seconds).

3. Desired interval between interim calculations (in seconds).

The output from the program is quite prompt.

1. Eccentricity of new orbit.

2. Length of new Semi-Latus Rectum in nautical miles.

3. Length of new radius vector in nautical miles.

4. New radius vector angle with respect to the <u>new</u> perigee axis.

5. Direction and amount (in degrees) of orbital rotation as measured between the two perigee axes.

Obviously, this program can be sequentially staged to move the ship about quite freely. Attempting to move under power from one polar plot to another is substantially challenging, particularly if velocities are to be matched. Repeated use of this program demonstrates that axial shifts are much larger for low eccentricity orbits than for ones of higher eccentricity given the same amount of disturbance through added thrust.

The general factors that must be taken into consideration are presented in crude and exaggerated form in the following figures.



Figure 1.

$$r_o = \frac{L_o}{1 + e_o \cos \theta_o}$$

The original orbital equation in polar form.



Figure 2.

Further detail at P_o showing inclination and burn angles.

P_o (r_o, θ_o)



Figure 3.

Rotation of orbit measured through Perigees.

An apology is extended for the profusion of factors in Figures 1 through 3, but each has its particular place in orbital navigation. Before proceeding to a fuller discussion of the program a small development of the use of the inclination angle is in order. Certain liberties have been taken in its derivation which do not appear in the mathematical text and could be puzzling to the serious student.

By assuming that each moment the motors fire is their last moment, and consequently at each moment the ship is in a calculable orbit, these intermediate orbits may be continuously calculated. The program uses an interval of 5 seconds but is adjustable to any figure one may desire. Obviously the smaller the interval the more accurate the calculations become. In an orbit of thousands if not tens of thousands of seconds in period an interval of 5 seconds closely approximates the actual value. Assuming the 5 second (or less) interval is a sufficiently close approximation of the actual value, the calculations for instantaneous angle of orbital inclination become rather straightforward.



Figure 4.

Angle of Inclination

Since we are substituting a small t for dt the angle swept out by the radius vector for small t is arbitrarily held to approximate that angle swept out for dt, i.e., $d(\theta + \theta^*)$. The radius vector, r, increases by dr across angle $d(\theta + \theta^*)$ and the differential arc length approximated by its tangent has length $rd(\theta + \theta^*)$. Assuming $d\theta$ then the "new" radius vector crosses the tangent to the curve at right angles. By taking the angle vertical to $\emptyset$ the construction becomes clearer as noted on the following page.



Figure 5.

$$\text{Tan } \phi = \frac{dr}{rd(\theta + \theta^*)} \text{ from use of vertical angles}$$

(NOTE: Error of small t in place of dt is proportional to the difference between the arc length of the orbital path and the length of the equivalent tangent segment.)

If $r = \dfrac{L}{1 + e\cos(\theta + \theta^*)}$ then $\dfrac{dr}{rd(\theta + \theta^*)} = \dfrac{e\sin(\theta + \theta^*)}{1 + e\cos(\theta + \theta^*)}$ by the use of the simple differential formulas. Since the angle of inclination describes the orbit all that remains is the task of fitting all the data and formulas together.

Since the companion program was to be as realistic as possible, an actual spaceship was modeled within the program. The ship has a certain mass, thrust, and fuel usage compatible with current chemically propelled rockets. These ship parameters can be varied for any number of reasons.

The ship is at P_o: (r_o, θ_o) in an originating orbit with known constants of:

$$r_o = \frac{L_o}{1 + e_o\cos\theta_o} \qquad \text{(E1)}$$

At this point the ship has a velocity, v_o, determined by

$$v_o^2 = Gm\left(\frac{2}{r_o} - \frac{(1 - e_o^2)}{L_o}\right) \qquad \text{(E2)}$$

where Gm is Earth gravitational constant.

Furthermore, the angle made by the tangent to the orbit and the perpendicular to the radius vector at P_o is the angle of inclination and is

$$\tan \phi_o = \frac{e_o \sin \theta_o}{1 + e_o \cos \theta_o} \qquad (3)$$

The ship is rotated to the burn angle (measured from the perpendicular to the radius vector to the axis through motors) and the motors are fired. At the end of (m) seconds the ship has added perpendicular and radial distance vectors due to this thrust. These distance vectors are:

$$x_1 = \tfrac{1}{2}m^2 \ \bar{a} \cos z \qquad (E4)$$

and

$$y_1 = \tfrac{1}{2}m^2 (\bar{a} \sin z) \qquad (E5)$$

The supporting equations for these vectors are:

$$g_o = \frac{Gm}{r_o^2} \qquad (E6)$$

$$\bar{a} = \frac{T\bar{g}}{(w - m \cdot k/2)} \qquad (E7)$$

$$h = mv_o \sin \phi_o + \tfrac{1}{2}m^2 \ (\bar{a} \sin z) \qquad (E8)$$

$$g_h = Gm/(r + h)^2 \qquad (E9)$$

and

$$\bar{g} = (g_o + g_h)/2 \qquad (E10)$$

where

t is the ship thrust in pounds;

w is the ship weight in pounds;

k is the pounds of fuel burned per second;

g_o is gravitational acceleration at P_o ;

g_h is gravitational acceleration at height *h*;

$\overline{g}$ is mean gravitational acceleration during time *m*;

and

$\overline{a}$ is mean acceleration from ship during time *m*.

To keep these equations at a secondary school level of complexity certain simplifying assumptions have been made. Evidently $\overline{g}$ and $\overline{a}$ are not true means but first approximations from their respective infinite series. Also, for *m*, a time interval of $m \leq 5$ should be used. The results of (E3) are based on the tangent to a curve closely approximating the curve for small distances; i.e., the differential arc length is extended.

At the end of *m* seconds of firing the ship is in a new orbit which may be computed in terms of the old orbital parameters.

For the point of entry $P(r, \theta)$ on the new orbit, *r* is computed as

$$r^2 = (x_1 + x_2)^2 + (r_o + y_1 + y_2)^2 \qquad \text{(E11)}$$

$$x_2 = mv_o \cos \phi_o \qquad \text{(E12)}$$

$$y_2 = mv_o \sin \phi_o \qquad \text{(E13)}$$

The radius vector has now moved forward angle (w) from its old position. Further orbital relationships are:

$$\theta = w + \theta_o \qquad \text{(E14)}$$

$$\text{Tan } w = (x_1 + x_2)/r \qquad \text{(E15)}$$

$$m^2v^2 = (x_1 + x_2)^2 + (y_1 + y_2)^2 \qquad \text{(E16)}$$

$$\text{Tan } \phi = (y_1 + y_2)/(x_1 + x_2) \qquad \text{(E17)}$$

The original reference line which was the perigee axis of the old orbit has rotated to reflect the perigee axis of the new orbit; i.e., the ship's reference line has rotated through an angle of θ^*.

The new orbit through point P is defined by:

$$r = \frac{L}{1 + e \cos(\theta + \theta^*)} \qquad \text{(E18)}$$

$$v^2 = Gm \left(\frac{2}{r} - \frac{(1 - e^2)}{L} \right) \qquad \text{(E19)}$$

$$\tan \phi = \frac{e \sin (\theta + \theta^*)}{1 + e \cos (\theta + \theta^*)} \qquad \text{(E20)}$$

The simultaneous solution of (E18-20), a tedious exercise in analytic geometry, yields:

$$e^2 - 1 = \frac{rv^2 \cos^2 \phi \, (rv^2 - 2Gm)}{Gm^2} \qquad \text{(E21)}$$

$$L = \frac{r^2 v^2 \cos^2 \phi}{Gm} \qquad \text{(E22)}$$

$$\sin \theta^* = -c \sin \theta \pm \cos \theta \sqrt{1 - c^2} \qquad \text{(E23)}$$

where

$$c = (L - r)/er = \cos (\theta + \theta^*) \qquad \text{(E24)}$$

A sufficient condition for $-1 \leq c \leq 1$ is that $P \leq r \leq A$ and of the two solutions to (E23) the sum is used in the first two quadrants while the difference is used in the last two quadrants.

XFR

```
100 PRINT "THIS PRØGRAM IS DIRECTLY KEYED TØ THE TØPIC ØUTLINE"
110 PRINT "'ØRBITAL TRANSFER' AND THE NUMBERS IN THE REM CØLUMN"
120 PRINT "ØF THE LIST ØF THIS PRØGRAM REFER TØ THE EQUATIØN NUMBERS"
130 PRINT"IN THE TEXT."
150 PRINT
160 LET T = 75000                          'THRUST
170 LET G = 62747                          'GM
180 LET W = 50000                          'WEIGHT
190 LET K = 100                            'FUEL USE
200 LET I = 180/3.14159265                 'RAD/DEG
210 LET J = 3.14159265/180                'DEG/RAD
220 PRINT "ECCENTRICITY";
230 INPUT E
240 PRINT "SEMILR";                       'N.M.
250 INPUT L
260 PRINT "PØSITIØN ANGLE";                'DEG
270 INPUT P
280 LET P = J * P
290 PRINT "BURN ANGLE";                    'DEG
300 INPUT Z
310 LET Z = J * Z
320 PRINT "BURN TIME";                     'SEC
330 INPUT T(1)
340 PRINT "INCREMENT";                     'SEC
350 INPUT M
360 FØR N = M TØ T(1) STEP M
370 LET R = L/(1+E*CØS(P))                                  '(E1)
390 LET V = SQR(G) *SQR((2/R)-((1-E↑2)/L))                  '(E2)
410 LET U = ATN(E*SIN(P)/(1+E*CØS(P)))                      '(E3)
430 LET G(1) = G/R↑2
450 LET A = T*G(1)/(W-M*K/2)                                '(E7)
470 LET H = M*V*SIN(U)+.5*M↑2*A*SIN(Z)                     '(E8)
490 LET G(2) = G/(R+H)↑2
510 LET G(3) = (G(1) +G(2))/2                               '(E10)
520 LET A = T*G(3)/(W-M*K/2)                                 '(E7)
540 LET X(1) = .5*M↑2*A*CØS(Z)                               '(E4)
560 LET Y(1) = .5*M↑2*A*SIN(Z)                               '(E5)
580 LET X(2) = M*V*CØS(U)                                   '(E12)
600 LET Y(2) = M*V*SIN(U)                                   '(E13)
620 LET P(1) = ATN((X(1)+X(2))/R)                                   '(E15)
640 LET R = SQR((X(1)+X(2))↑2+(R+Y(1)+Y(2))↑2)                      '(E11)
660 LET P = P(1)+P                                          '(E14)
680 LET S = SQR((X(1)+X(2))↑2+(Y(1)+Y(2))↑2)                 '(E16)
690 LET V = S/M
710 LET U = ATN((Y(1)+Y(2))/(X(1)+X(2)))                    '(E17)
730 LET E =SQR((R*V↑2*CØS(U)↑2*(R*V↑2-2*G)+G↑2)/G↑2)
750 LET L = ((R*V*CØS(U))↑2)/G                              '(E22)
770 LET C = (L-R)/(E*R)                                      '(E24)
780 IF P < 6.28318530 THEN 800
790 LET P = P-6.28318530
```

XFR (CONTINUED)

```
800 IF P > 3.14159625 THEN 840
810 LET S = (-C*SIN(P)+COS(P)*SQR(1-C↑2))/2
811 LET S=2*S
820 LET S = ATN(S/SQR(1-S↑2))
830 GOTO 930
840 LET S = (-C*SIN(P)-COS(P)*SQR(1-C↑2))/2
841 LET S=2*S
850 LET S = ATN(S/SQR(1-S↑2))
860 IF Y(1) < 0 THEN 890
870 LET P = P+S
875 LET S(1) = S(1) - S
880 GOTO 970
890 LET P = P-S
895 LET S(1) = S(1)+S
900 LET S = ATN(D(2)/SQR(1-D(2)↑2))                    'AXIAL SHIFT
910 GOTO 970
920
930 IF Y(1) < 0 THEN 960
940 LET P = P-S
945 LET S(1) = S(1) + S
950 GOTO 970
960 LET P = P+S
965 LET S(1) = S(1)-S
970 LET W = W-M*K/2
980 IF P < 6.28318530 THEN 1000
1000 LET W = W-M*K/2
1020 NEXT N
1080 PRINT
1090 PRINT "ECCENTRICITY","SEMI LR","RAD V","RAD ANGLE","AXIAL SHIFT"
1100 PRINT E,L,R,P*I,S(1)*I
1500 END
```

```
RUN

XFR        03/18/70  09:32

THIS PRØGRAM IS DIRECTLY KEYED TØ THE TØPIC ØUTLINE
'ØRBITAL TRANSFER' AND THE NUMBERS IN THE REM CØLUMN
ØF THE LIST ØF THIS PRØGRAM REFER TØ THE EQUATIØN NUMBERS
IN THE TEXT.

ECCENTRICITY? .01
SEMILR? 4800
PØSITIØN ANGLE? 90
BURN ANGLE? 30
BURN TIME? 20
INCREMENT? 5

ECCENTRICITY    SEMI LR        RAD V          RAD ANGLE      AXIAL SHIFT
 7.49169 E-3     4896.69        4924.29        184.512        -93.667

TIME: 0.498 SEC.
READY
```